
Doing graphics

with Visual DataFlex
version 1.4

by Sture Andersen

The packages in the VdfGraph upload were programmed by me and are to be considered 'public domain' and may be distributed freely as is or as part of a product including these files with no further permission than this statement.

This document describes the public domain package file called VDFGRAPH.UTL. The classes defined in there provide the ability to draw on screen. The method used does not require any external DLL’s or OCX’s. Only the ability of VDF to perform external function calls into the standard Windows GDI has been utilized.

The knowledge necessary to write the initial versions of the classes was obtained by reading Programming Windows 95 by Charles Petzold (MS Press) and by studying the dfshape.pkg file supplied with VDF. Initially I was impressed and inspired by a DataFlex 3.05 for Windows application designed by Klaus Møller Jensen at Danbrew for Carlsberg.

The source code reproduced in this document and the source code of the samples uses a set of classes called APS. APS is a set of classes that will size and locate themselves automatically. The graphic classes however, are not depending upon APS to function. APS is used because it significantly reduces the number of lines in the source code (documentation available if you e-mail me at the address below).

Inspired by Geoff Furlong of MSG Systems competent interest for it I thought that I would encourage others to also expand its functionality if they feel capable and motivated. How about it? User-driven public domain packages! Provided of course that I stay the editor and decide what goes in and what doesn't, I am prepared to discuss anything. Anyway, to this end I will try to explain the internal workings of the most important class (the GraphicArea class) in a separate chapter.

Warning: This is not a user friendly tool that let’s you point to tables and fields in order to automatically generate some flashy graphic report. The classes defined in this package have not been integrated with the IDE (see instructions later in this document). And there is no other visual design tool available to help you generate graphics. This is all code based and you will have to do all the work in your editor. Evaluation should begin by compiling the gr.src sample program.

On the other hand: It's pretty fast, no problems installing OCX's on ill tempered NT workstations and you are in control of your tool. And it still produces nice looking graphics with a little enthusiasm put into it.

If you have any questions please use Data Access public news groups (news://dataaccess.com/dac-public-newsgroups.visual-dataflex-support) and put VDFGRAPH in the subject line.

November 9th, 1998

Sture Andersen

Introduction

All graphics done with this utility is drawn within an object of the class GraphicArea. Such an object is capable of drawing lines, circles, boxes and text. This is done in a virtual coordinate system of range 0 to 10000 in both dimensions. (0,0) is the upper left corner. The object takes care of all housekeeping required to make Windows draw the desired results on screen. That is all this object is capable of. Drawing a bar chart using only the services of this class would be a tedious affair.

Therefore an object of class cCoordinateSystem may be put inside a GraphicArea object. This object is capable of drawing a coordinate system within a specified part of the GraphicArea including measures along the axis. It is also capable of translating between its own coordinate system and the coordinate system of the GraphicArea.

If we wanted to do a bar chart it would still be pretty complicated even if the cCoordinateSystem class makes it possible to paint using the “right” coordinate system. We therefore insert an object of class cBarChart inside the cCoordinateSystem object. In there we talk in terms of columns, series and stacks.

Class: GraphicArea

The graphics above was generated with an object of class GraphicArea. The structure of the source code for this view would be:

use VdfGraph.utl

activate_view run_test1 for oGraph_Example1

object oGraph_Example1 is a View

  set size to ...

  set label to "Example"

  object oGraph is a GraphicArea
    set location to 5 5

    set size to 200 200

    procedure draw_data
      forward send draw_data
      -- do the drawing here --

    end_procedure

    send draw_data
  end_object

  object oBtn is a Button

    set size to ...

    set location to ...

    on_item "Close" send close_panel

  end_object

end_object
The complete source code for the above image may be inspected in the file grtest1.pkg of the graphic samples. 

In order to make an object of this class display anything you must tell it what to display(!). The most important method of the GraphicArea XE "aps.GraphicArea"  class is the draw_data procedure (no parameters). This message must be sent (and forwarded) to the object. Let’s move to something slightly less complex:

The draw_data procedure of this view looks like this:

procedure draw_data

  forward send draw_data // Must be forwarded (first thing)

  // Draw yellow rectangle:

  send SetFillColor clYellow

  send SetPenColor  clBlack

  send AddRectangle 1000 1000 7000 5000

  // Print centered text (in blue):

  send SetStockFont SYSTEM_FONT

  send SetTextAlign VDFGR_TA_CENTER

  send SetTextColor clBlue

  send AddText      "Hello World" 4000 7500

  // Draw solid red line:

  send SetPenColor  clRed

  send SetPenStyle  PS_SOLID // Solid pen

  send SetPenWidth  2

  send AddLineMvTo  8500 1000

  send AddLineGoTo  8500 9000

end_procedure
Note that the coordinate system defined by the GraphicArea class is virtual. No matter what the visual size has been set to the visual range of both dimensions will be 0 to 10000.

The following drawing messages may be used inside the draw_data procedure:

· procedure SetPenColor integer color#
· procedure SetPenWidth integer width#

· procedure SetPenStyle integer style# (only works when pen width has been set to 1 or 0)
· procedure SetFillColor integer color#
· procedure SetBackColor integer color#
· procedure SetTextAlign integer align#
· procedure SetTextColor integer color#

· procedure SetStockFont integer font#
· procedure SetTTFont string name integer sz integer angle bool bold bool ital bool und
· procedure SetDotSize integer size#
· procedure SetDotStyle integer style#
· procedure SetDotAlign integer align#
· procedure SetRoundRectFactor integer factor# (not interesting)
· procedure AddRectangle integer x1 integer y1 integer x2 integer y2
· procedure AddEllipse integer x1 integer y1 integer x2 integer y2
· procedure AddRoundRect integer x1 integer y1 integer x2 integer y2
· procedure AddLineMvTo integer x integer y
· procedure AddLineGoTo integer x integer y
· procedure AddText string str# integer x integer y
· procedure AddPolyPoint integer x integer y
· procedure AddPolyLine
· procedure AddPolyGon
· procedure AddDot integer x integer y
The function of all "set"-procedures will remain in effect until explicitly changed.

Parameter color# may be any of the following (defined in standard DAC package rgb.pkg):

clAqua, clBlack, clBlue, clDkGray, clFuchsia, clGray, clGreen, clLime, clLtGray, clMaroon, clNavy, clOlive, clPurple, clRed, clSilver, clTeal, clWhite, clYellow or indeed any RGB value.

Parameter font# (as used with the SetStockFont message) may be:

· SYSTEM_FIXED_FONT

· OEM_FIXED_FONT

· ANSI_FIXED_FONT

· ANSI_VAR_FONT,

· SYSTEM_FONT

· DEVICE_DEFAULT_FONT

· SYSTEM_FIXED_FONT

Parameter align# may be:

· VDFGR_TA_LEFT

· VDFGR_TA_CENTER

· VDFGR_TA_RIGHT

· VDFGR_TA_TOP 

· VDFGR_TA_VCENTER

· VDFGR_TA_BOTTOM

You may change the background color of the graphic area by setting the value of property pBackColor to any RGB value.

A note on texts

VDFGRAPH handles two types of fonts: Stock fonts (message SetStockFont) and True Type fonts (message SetTTFont).

Until I did VDFGRAPH I never knew what a stock font was. It's a kind of system font. There are six of them and you can see them all in one of the test views. They cannot be scaled and they cannot be rotated. On the other hand, they are more simple to administrate and therefore probably faster than TT fonts.

A TT font, on the other hand must be created before it can be selected (you do not create stock fonts, you just select them). In fact, you don’t do anything in either case. VDFGRAPH does it for you.

Dots

Dots are small markers used primarily when doing graphs. In addition to fill color and pen color dots have a size and a style. The following styles are available:

· DT_PIXEL

· DT_CROSS

· DT_PLUS

· DT_HORIZONTAL

· DT_VERTICAL

· DT_CIRCLE

· DT_TRIANGLE_UP

· DT_TRIANGLE_DOWN

· DT_TRIANGLE_RIGHT

· DT_TRIANGLE_LEFT

· DT_SQUARE

· DT_DIAMOND

In this order they appear below. The corresponding dot size is indicated in the header line:

Unlike rectangles and everything else (except texts) dots do not scale according to the size of the graphic area inside which they are painted. Note also that the above sample was generated while pen color was set to black and fill color was set to red. Still you would not say that all the dots are red. So, for some dot styles the color of the dot is determined by the pen color and for others by the fill color.

Per default dots appear centered vertically and horizontally, but this may be changed with the SetDotAlign message. This procedure takes one parameter that may be any of the following:

· VDFGR_TA_LEFT

· VDFGR_TA_CENTER

· VDFGR_TA_RIGHT

· VDFGR_TA_TOP 

· VDFGR_TA_VCENTER

· VDFGR_TA_BOTTOM

The default dot alignment mode is (VDFGR_TA_CENTER+VDFGR_TA_VCENTER). 

The draw_data procedure

In a minute we will be putting graphical objects inside the graphical area. It is important to know that these all understand the message draw_data. Yet the message may only be sent to the graphical area itself. The graphical area object will send draw_data to all its children, which is part of the reason that it is very important that the message is forwarded when augmented.

Resizing

In the sample application you will see examples of re-sizable graphic areas implemented by the use of APS. It does not matter what re-sizing strategy you are using (refer to Mastering Visual DataFlex for an article on that). The graphic area will repaint correctly if resized, no matter how. 

Class: cCoordinateSystem
In order to do a graph or a bar chart you will need a coordinate system that must be defined inside the graphic area. This will be of class cCoordinateSystem XE "cCoordinateSystem" . Such an object knows how to auto scale the axis, draw text on the axis and how to convert between the coordinates of the coordinate system and the graphic area.

All the drawing messages of the GraphicArea class are also understood by the cCoordinateSystem class. The difference is of course that all coordinate parameters are interpreted in their own coordinate system.

The first thing you need to tell the coordinate system is its size and location within the graphic area:

set area_location to 2000 2000

set area_size to 6000 7000
This is done in the coordinate system of the graphic area. Remember that the coordinate range is always 0 to 10000 in both dimensions regardless its GUI-size.

You may define as many coordinate systems inside a graphic area as you wish. You may tile or overlap these as needed by setting the above properties. 

Scaling the axis

Secondly you have to provide information on the range of the scales in the X and Y directions. For each axis you have two options. Either you specifically set the range of the axis or you may set it indirectly using the auto scale capability. You tell the object which method you will use by setting the value of the pyAutoScale property to true or false (the corresponding property for controlling X-axis behavior is called pxAutoScale).

Let us say that you want to set the range of the Y-axis manually. You must set pyAutoScale to false. In order to specify the range you specify the lower limit, a number of steps and the step-size. This will tell the Y-axis to cover the range from 80 to 280 in ten steps:

set pyAutoScale to false

set pyLowValue  to 80

set pySteps     to 10

set pyStepSize  to 20
If on the other hand you would like to take advantage of auto scaling you would write:

set pyAutoScale to true

set pyMinValue to 80

set pyMaxValue to 280 

set pyMinSteps to 10

set pyMaxSteps to 10
The auto scaling mechanism selects the most efficient number of steps to cover the range indicated by the settings of the pyMinValue and pyMaxValue properties. It will select a number of steps between pyMinSteps and pyMaxSteps. Setting both to 10 forces the auto-scaler to select 10 steps.

Two more properties allow you to control the auto scaling. If you want 0 to be included into the scale regardless of the pyMinValue and pyMaxValue properties you must set pyZeroBased to true. The second is called pyAirPct. Setting this to 5 will ensure at least 2.5% space above pyMaxValue and 2.5% below pyMinValue. An identical set of properties exists to control the X-axis.

Once the scaling of the axis has been determined you may use these functions to obtain the range and the highest value covered by each of the axis.

· function nxHighValue returns number

· function nyHighValue returns number

· function nxRange returns number

· function nyRange returns number
Adding texts

For each major step of the axis you may specify a text to print. The default behavior however, is to print the numeric value corresponding to each major step. This behavior is controlled by the value of properties pxAutoAxisText and pyAutoAxisText. These may take on three values:

· AT_NONE
No text on axis

· AT_AUTO
Numeric values 

· AT_TEXT
Manually specified texts

If AT_TEXT is used you may specify a text per major step using the properties AxisTextX and AxisTextY.

For example:

Set pxAutoAxisText to AT_TEXT

Set AxisTextX 0 to "Jan"

Set AxisTextX 1 to "Feb"

Set AxisTextX 2 to "Mar"

Set AxisTextX 3 to "Apr"

Set AxisTextX 4 to "May"

Set AxisTextX 5 to "Jun"
If you do the above you probably do not want the texts to appear on the major step values but rather in between them. This behavior is controlled by the pBarChartState property. If you set this to true, the text will move half a major step up the axis. (This property is automatically set to true if you put a cBarChart object inside the coordinate system). In addition to this you may set a title for each of the axis: pTitleX and pTitleY.

Now, the axis texts will print just outside the coordinate system as expected. How much outside may be controlled via the pxTextOffSet and pyTextOffSet properties. The value of these must be set in the measures of the surrounding graphics area and their default values are 50.

Visual appearance

Background color of the coordinate system

Setting the pColor property changes the color of the area covered by the coordinate system. The default setting (-1) makes it transparent.

Appearance of the grid
You may specify that a grid should be drawn to make the coordinate system visible. With respect to this grid we talk about the major scale and the minor scale in both dimensions. The major scale is that indicated by the number of steps (refer to the auto scale section).  That is, if you have 6 steps on the Y-axis, the major grid of that dimension will have 7 lines to it.

The minor scale is a subdivision of the major scale. For instance, you may specify that for each major step you want 4 minor steps.

Set each of the properties pxMajorScale and pxMinorScale to one of the following values:

· GLS_NONE
Grid line not visible

· GLS_LINE
Solid line

· GLS_DOT
Dotted line

Set pxMinorDiv to the number of minor steps per major step.

An identical set of properties exists for the Y-axis. (pyMajorScal, pyMinorScale and pyMinorDiv).

To set the color of the grid lines set properties pMinorScaleColor and pMajorScaleColor to a legal RGB value. These control the color of the grid lines in both dimensions.

The cCoordinateSystem class is used in the grtest02.pkg file.

Class cBarChartData
Put an object of this class inside a cCoordinateSystem XE "cCoordinateSystem"  object to display a bar chart. Before getting further into this we need to get some names defined. This dialog is defined in grtest8.pkg:

This bar chart is made up of 5 columns. Each column is made out of three series (red, blue. green). Each series within a column consists of two stacked bars (two stacks). The values are indicated to the left of the chart.

The forms labeled Min: and Max: are used to control the random generating of test data. It means that all data items will be between Min and Max. If this interval is changed to -12 to 12 you will get a chart like this:


In both cases a number represents each rectangular colored area. If the area is below the y=0 line this number is negative.

Adding data

Before we can begin to add data to the bar chart it needs to know the number of stacks and series. These are stored in properties pStacks and pSeries. Their default value is 1.

· property integer pStacks
· property integer pSeries
Once these have been set we should make sure that data does not exist already:

· Procedure Reset_Data
Now we can use the following messages to put data in the bar chart.

· Procedure Sto_Data number value# integer column# integer serie# integer stack#
· Procedure Sum_Data number value# integer column# integer serie# integer stack#
Procedure Sto_Data stores the value# of the specified data item while Sum_Data adds it to whatever it might already be.

· Function nRcl_Data.iii integer column# integer serie# integer stack# returns number

This function may be used to retrieve the value of a specified data item.(Rcl=ReCall).

· Function iColumns returns integer

This function returns the number of columns used in the bar chart.

· Function nColumnSum integer column# integer serie# integer code#  returns number

Returns the sum of the data items in a single bar (specified column and series). Parameter code# may take on the following values:

· 0:
Sum of the positive data items

· 1:
Sum of the negative data items

· 2:
Sum of all data items

Collecting data

Graphics in general may be used in a variety of ways. It may be that for each record you display in a conventional view you have a graphic representation of some information related to that record.

There is nothing in the VdfGraph.utl package file to help you actually scan your data files and collect data for displaying. However, the use of graphics may be divided into the following groups:

This paragraph is obviously not finished. So for now, you have to figure this part out yourself (as others before you). Sorry.

Printing graphics

A German company called Ideal Software makes a print engine called 'Virtual Print Engine' or VPE for short. This engine may be used from any programming language that supports calling external DLL's.

VPE is faster and more precise than WinPrint. Unfortunately it is rather expensive. 550$ per developers seat, distribution is free (same principle as VDF).

If you are the happy owner of a VPE license you may print the contents of a graphic area by sending the message print_graphic_area. The procedure takes five parameters, the first being the object id of the graphical area that should be printed. The latter four arguments specifies the top left and the bottom right corner (in 1/10th’s of a millimeter) of the area on the paper. The package file VPEGRAPH.UTL provides this feature.

Note that VDFGRAPH does not produce WYSIWYG graphics on printer. The font handling is too different (and I have broken too many rules) to attach the above label to it. It is however possible to make special versions of your screen graphics that prints high quality graphics on printer.

Even if you are not currently using VPE, you may still try it out by getting a fully functional DEMO version from the web site of Ideal Software:

· Go to www.IdealSoftware.com

· Download the 32 bit trial version of VPE

· Put VPE32.DLL in the WINDOWS directory

It is not possible to print this kind of graphics with WinPrint. Somebody published information (DBoard on www.datrixnet.com) on how to achieve better XY control of WinPrint, so maybe now it could be done.

A note on the demo program

The grdemo.src is a program for demoing the capabilities of the VdfGraph.utl package file. The samples are hand coded with a set of classes called APS (aps.pkg) that will help minimize the number of source code lines needed. I hope you are able to read the sample code anyway and gain the knowledge needed to do the graphics you want.

However, you do not need APS to do graphics. You will just have to use the GraphicArea class instead of the aps.GraphArea class used in the examples.

How does it work?

One of the reasons that I did the VDFGRAPH package was to provide encapsulation for as much of the housekeeping as possible without limiting what can be done with the Windows GDI. To the use of each of the interfaced functions there is connected an amount of knowledge of how to represent Windows data structures in DataFlex and so forth. It is good to have this knowledge packed away in a central place ready for use. This central place is the GraphicArea class.

Obviously not all functionality of the Windows GDI is interfaced but the current approach is open to enhancements.

I have already once experienced that somebody sat down and figured out how it all really works (Geoff Furlong), only he did it without any documentation or help. But because he did that, we were able to add the ability to draw poly-lines and polygons.

I thought that maybe I could get more people on the move by describing the mechanisms of the GraphicArea class. All the other classes really just utilize the services of this one class.

Though the source code for the GraphicArea class may look complex and strangely structured, the original thought really is quite simple.

Design goal

Part of the Windows 95 operating system is a collection of functions that will draw stuff on the screen. This part is referred to as the Windows GDI. GDI is short for Graphical Device Interface. Such a function is reached from within Visual DataFlex by declaring an external function. 

The Windows GDI function library is located in the Windows system directory as a DLL called gdi32.dll. To interface some of the functions in there, the standard VDF package wingdi.pkg declares a number of these external functions. Here is one:

External_Function32 Rectangle "Rectangle" GDI32.dll ;

Handle hDC Integer X1 Integer Y1 Integer X2 Integer Y2 Returns Integer
This statement creates a Visual DataFlex function called Rectangle that will execute a function of the same name present in the DLL file called gdi32.dll. It takes 5 parameters and returns an integer. Common to nearly all the external functions relevant to drawing graphics is that their return value is of no interest. We do not care what (if anything) it returns.

The parameters X1, Y1, X2, Y2 signifies the top left and the bottom right corner of the rectangle we want to draw. The first parameter is called hDC and is of type handle. If we want to draw a rectangle Windows has to know in which window we want to draw it. More precisely, Windows has to know in which device context we want to draw the rectangle. The frame color and the fill color must be set via some other external function prior to calling the Rectangle function.

The logic that obtains a handle for a device context may be studied in the standard VDF package called dfshape.pkg. A sample usage is presented in the VDF help file (class: ShapeControl).

Using a number of external functions we may paint a lot of lines, rectangles, circles and texts within our device context. Unfortunately our task does not end there. Windows does not “remember” what is painted. The first time our drawing should re-display after having been minimized, covered by another window or having been moved outside the display area it will re-display as blank. We have to draw it all again.

Now, the graphics that we draw may be the result of running through a very large database. We do not want to do that again in order to re-display. We have to keep just enough information to re-draw our graphics without actually re-generating it.

Approach

To meet the demand of being able to re-draw the entire graphic area, we save all the drawing information in an array object. The idea is that all the Set- and Add- messages we send to the graphic area are captured and stored in the array together with their parameters. When the GraphicArea object actually paints anything on screen it is really just “emptying” the contents of the array by executing our previously given instructions sequentially as stored in the array.

Embedded in a GraphicArea object is an array called program_RAM. This is the array to which we write all the instructions and its name indicates the metaphor I thought of when I made it. I think of it as the memory of a simple computer. The instruction set of this computer is the set of Set- and Add- messages.

In order to make everything happen as fast as possible I based the method on intensive use of global variables and good old-fashioned subroutines (GOSUB).

Let’s see what happens when we use a GraphicArea object to draw a rectangle, a line and a text. Here is the example of a previous section:

 1 procedure draw_data

 2   forward send draw_data // Must be forwarded

 3                  

 4   // Draw yellow rectangle:

 5   send SetFillColor clYellow

 6   send SetPenColor  clBlack

 7   send AddRectangle 1000 1000 7000 5000

 8                

 9   // Print centered text (in blue):

10   send SetStockFont SYSTEM_FONT

11   send SetTextAlign VDFGR_TA_CENTER

12   send SetTextColor clBlue

13   send AddText      "Hello World" 4000 7500

14                

15   // Draw solid red line:

16   send SetPenColor  clRed

17   send SetPenStyle  PS_SOLID // Solid pen

18   send SetPenWidth  2

19   send AddLineMvTo  8500 1000

20   send AddLineGoTo  8500 9000

21 end_procedure
What happens when this procedure executes? Remember that this is the process of filling up the program_RAM object rather than actually painting anything on the screen (yet).

First thing that happens is that the message is forwarded. Thereby the following piece of code executes:

move (program_RAM(current_object)) to gr$CPU$RAM // Global integer

send delete_data to gr$CPU$RAM

move 0 to gr$CPU$PC // Also a global integer
At this point you have to know that two global integers (with rather odd names) have been declared called gr$CPU$RAM and gr$CPU$PC. While the draw_data procedure executes the gr$CPU$RAM variable contains the object ID of the program_RAM object. This gives a somewhat faster access to our “memory” object. The other variable gr$CPU$PC plays the role of a program counter when we eventually will “run” our graphical “program”. But we also use it during the recording process. During the recording process gr$CPU$PC always points to the next available item in the program_RAM array. And so we start by setting it to 0.

So, forwarding the draw_data message resets the program_RAM object and all these global variables.

Before I go any further I have to briefly explain the notion of procedure identifiers. In DataFlex these two pieces of code does the same thing:

send Request_Save

is equivalent to:

integer msg# // Integer that will hold a message identifier

move msg_Request_Save to msg#

send msg#

When a Request_Save procedure is defined the compiler adds msg_Request_Save to its symbol table (and not Request_Save). The msg_Request_Symbol is replaced to an integer. Therefore in theory one could also write send 375 (or whatever) instead of send Request_Save. OK, you’ve got it! The point is that you can store a number of procedure identifiers in an array and execute them sequentially.

Now, line 5 of the example reads

send SetFillColor clYellow

We call a procedure called SetFillColor with one argument, which is a constant defined in standard DAC package rgb.pkg. If you look in VdfGraph.utl you will see a very strange looking definition of this procedure. All you will see is this:

vdfgraph$Procedure_OneArg SetFillColor msg_mthd_SetFillColor

What the hell is that, I’d expect you to think. Well, I have been programming DF for many years and when I can see that I am going to have to write a lot of very similar procedures, I just have to find a way to get around it easily. 

So, in the same file I have defined a command by the name of vdfgraph$Procedure_OneArg. Its ugly name indicates that I do not ever expect to ever use it outside this package. It is defined like this:

#COMMAND vdfgraph$Procedure_OneArg

  procedure !1 integer int#

    set value of gr$CPU$RAM item gr$CPU$PC to !2

    set value of gr$CPU$RAM item (gr$CPU$PC+1) to int#

    move (gr$CPU$PC+2) to gr$CPU$PC

  end_procedure

#ENDCOMMAND
A command it is called, but a macro it is. When used like above it expands to this:

procedure SetFillColor integer int#

  set value of gr$CPU$RAM item gr$CPU$PC to msg_mtdh_SetFillColor
  set value of gr$CPU$RAM item (gr$CPU$PC+1) to int#

  move (gr$CPU$PC+2) to gr$CPU$PC

end_procedure

What you see here is that calling the SetFillColor procedure will add two items to the program RAM object. The first item added is the identifier of another procedure called mtdt_SetFillColor (the compiler prefixes procedure names with a msg_, and I do it to get the identifier of the procedure) and the second item added is the argument we passed to the SetFillColor procedure (clYellow). Finally the program counter is incremented by two to keep it pointing to the next available item in the program RAM array.

Line 6 is:

send SetPenColor clBlack

Same kind of story. Two items are added to the program RAM object, the first being the identifier of a procedure (msg_mthd_SetPenColor) and the second being the constant clBlack.

Line 7:

send AddRectangle 1000 1000 7000 5000

This really is very similar to what happened in lines 5 and 6 but I’ll just go through it again. When looking in the package file you can see that there is no conventional definition of a procedure called AddRectangle. What you do find is this line:

vdfgraph$Procedure_PlaneArg AddRectangle msg_mthd_Rectangle

that when paired with this macro definition:

#COMMAND vdfgraph$Procedure_PlaneArg

  procedure !1 integer x1# integer y1# integer x2# integer y2#

    set value of gr$CPU$RAM item gr$CPU$PC to !2

    set value of gr$CPU$RAM item (gr$CPU$PC+1) to (x1#*65536+y1#)

    set value of gr$CPU$RAM item (gr$CPU$PC+2) to (x2#*65536+y2#)

    move (gr$CPU$PC+3) to gr$CPU$PC

  end_procedure

#ENDCOMMAND
expands to this code:

procedure AddRectangle integer x1# integer y1# integer x2# integer y2#

  set value of gr$CPU$RAM item gr$CPU$PC to msg_mthd_AddRectangle
  set value of gr$CPU$RAM item (gr$CPU$PC+1) to (x1#*65536+y1#)

  set value of gr$CPU$RAM item (gr$CPU$PC+2) to (x2#*65536+y2#)

  move (gr$CPU$PC+3) to gr$CPU$PC

end_procedure

When we call the AddRectangle procedure in line 7 we end up adding three items to the program RAM array. The first item is an identifier for a procedure called mtdh_AddRectangle. In order to obtain the identifier for a procedure we just add msg_ in front of its name thereby making the compiler able to identify the symbol. The two remaining items are packed versions (the coordinate x,y is packed into one integer by this formula: x*65536+y) of the coordinates we passed to the AddRectangle procedure.

Lines 10 through 20 do the same thing all the way. Each adds a procedure identifier and some arguments to the program RAM object. After having executed line 20 the contents of the program RAM array will look like this:

	Item
	Contents
	Remarks

	0
	msg_mthd_SetFillColor
	Procedure ID

	1
	ClYellow
	Constant

	2
	msg_mthd_SetPenColor
	Procedure ID

	3
	ClBlack
	Constant

	4
	msg_mthd_AddRectangle
	Procedure ID

	5
	65537000
	1000*65636+1000

	6
	458573000
	7000*65536+1000

	7
	msg_mthd_SetStockFont
	Procedure ID

	8
	SYSTEM_FONT
	Constant

	9
	msg_mthd_SetTextAlign
	Procedure ID

	10
	VDFGR_TA_CENTER
	Constant

	11
	msg_mthd_SetTextColor
	Procedure ID

	12
	ClBlue
	Constant

	13
	msg_mthd_TextOut
	Procedure ID

	14
	“Hello World”
	Constant

	15
	252151500
	4000*65536+7500

	16
	msg_mthd_SetPenColor
	Procedure ID

	17
	ClRed
	Constant

	18
	msg_mthd_SetPenStyle
	Procedure ID

	19
	PS_SOLID
	Constant

	20
	msg_mthd_SetPenWidth
	Procedure ID

	21
	2
	Constant

	22
	msg_mthd_AddLineMvTo
	Procedure ID

	23
	557057000
	8500*65536+1000

	24
	msg_mthd_AddLineGoTo
	Procedure ID

	25
	557083000
	8500*65536+9000


Now we have filled an array with a lot of values but we haven’t drawn anything on the screen yet. The procedures that do the actual drawing are the ones whose identifiers we have stored in the array.

Before we can begin to “execute” the array we have to know about some more global variables. Here is a bunch of global integers:

· gr$PenColor

· gr$PenWidth

· gr$PenStyle

· gr$HatchStyle

· gr$FillColor

· gr$RoundRectFactor

· gr$PolyGonFillMode
These are state variables. When you draw a rectangle for instance, you do not specify the color of the frame or the inside of it. These colors are implicitly given by the values of the state variables gr$PenColor and gr$FillColor. Everything you draw that uses the “state” of the pen color – and that is virtually everything –will be the same color until you change the pen color state with the mthd_SetPenColor message.

When one of these state variables change, we have to inform Windows about it. There is quite a lot of housekeeping involved in doing that. Basically, when we ask Windows to paint anything we are selecting pens and brushes and when we change their properties we have to do things like de-selecting them, deleting them, creating new ones and selecting them. I recommend reading Programming Windows 95 by Charles Petzold from MS Press. Chapter 4 is what this is all about. Don’t be scared that all code is C++. It translates pretty well into Visual DataFlex once you get the hang of external functions.

Meanwhile back at my original point. Two more global variables are used to administrate this circus. To avoid over-informing Windows of changes I use integer variables gr$PenDirty and gr$BrushDirty to keep track of when the relevant state variables have changed. Only when I actually have to use the pen or the brush I inform Windows if there has been any changes since the last time I used them.

The details of the above explanation changes for each new feature I add, but you get the basic idea.

To be continued…

